Rules for integrands of the form $\left(d + e\, x^2\right)^p\, (a + b\, \text{ArcSinh}[c\, x])^n$

1. $\int \left(d + e\, x^2\right)^p\, (a + b\, \text{ArcSinh}[c\, x])^n\, dx$ when $e == c^2\, d$

   1. $\int \frac{(a + b\, \text{ArcSinh}[c\, x])^n}{\sqrt{d + e\, x^2}}\, dx$ when $e == c^2\, d$

      1. $\int \frac{(a + b\, \text{ArcSinh}[c\, x])^n}{\sqrt{d + e\, x^2}}\, dx$ when $e == c^2\, d \wedge d > 0$

         x: $\int \frac{(a + b\, \text{ArcSinh}[c\, x])^n}{\sqrt{d + e\, x^2}}\, dx$ when $e == c^2\, d \wedge d > 0$

- Derivation: Integration by substitution
- Basis: If $e == c^2\, d \wedge d > 0$, then $\frac{F[\text{ArcSinh}[c\, x]]}{\sqrt{d+e\, x^2}} == \frac{1}{c\, \sqrt{d}}\, \text{Subst}[F[x],\, x,\, \text{ArcSinh}[c\, x]]\, \partial_x \text{ArcSinh}[c\, x]$
- Rule: If $e == c^2\, d \wedge d > 0$, then

$$\int \frac{(a + b\, \text{ArcSinh}[c\, x])^n}{\sqrt{d + e\, x^2}}\, dx \longrightarrow \frac{1}{c\, \sqrt{d}}\, \text{Subst}\left[\int (a + b\, x)^n\, dx,\, x,\, \text{ArcSinh}[c\, x]\right]$$

- Basis: If $e1 == c\, d1 \wedge e2 == -c\, d2 \wedge d1 > 0 \wedge d2 < 0$, then $\frac{F[\text{ArcCosh}[c\, x]]}{\sqrt{d1+e1\, x}\, \sqrt{d2+e2\, x}} == \frac{1}{c\, \sqrt{-d1\, d2}}\, \text{Subst}[F[x],\, x,\, \text{ArcCosh}[c\, x]]\, \partial_x \text{ArcCosh}[c\, x]$
- Rule: If $e1 == c\, d1 \wedge e2 == -c\, d2 \wedge d1 > 0 \wedge d2 < 0$, then

$$\int \frac{(a + b\, \text{ArcCosh}[c\, x])^n}{\sqrt{d1 + e1\, x}\, \sqrt{d2 + e2\, x}}\, dx \longrightarrow \frac{1}{c\, \sqrt{-d1\, d2}}\, \text{Subst}\left[\int (a + b\, x)^n\, dx,\, x,\, \text{ArcCosh}[c\, x]\right]$$

- Program code:

```
(* Int[(a_.+b_.*ArcSinh[c_.*x_])^n_./Sqrt[d_+e_.*x_^2],x_Symbol] :=
  1/(c*Sqrt[d])*Subst[Int[(a+b*x)^n,x],x,ArcSinh[c*x]] /;
FreeQ[{a,b,c,d,e,n},x] && EqQ[e,c^2*d] && GtQ[d,0] *)
```

```
(* Int[(a_.+b_.*ArcCosh[c_.*x_])^n_./(Sqrt[d1_+e1_.*x_]*Sqrt[d2_+e2_.*x_]),x_Symbol] :=
  1/(c*Sqrt[-d1*d2])*Subst[Int[(a+b*x)^n,x],x,ArcCosh[c*x]] /;
FreeQ[{a,b,c,d1,e1,d2,e2,n},x] && EqQ[e1,c*d1] && EqQ[e2,-c*d2] && GtQ[d1,0] && LtQ[d2,0] *)
```

1: $\int \frac{1}{\sqrt{d+e x^2}\,(a+b\,\text{ArcSinh}[c\,x])}\,dx$ when $e == c^2 d \wedge d > 0$

- **Derivation: Integration by substitution**
- **Rule: If $e == c^2 d \wedge d > 0$, then**

$$\int \frac{1}{\sqrt{d+e x^2}\,(a+b\,\text{ArcSinh}[c\,x])}\,dx \longrightarrow \frac{\text{Log}[a+b\,\text{ArcSinh}[c\,x]]}{b\,c\,\sqrt{d}}$$

Program code:

```
Int[1/(Sqrt[d_+e_.*x_^2]*(a_.+b_.*ArcSinh[c_.*x_])),x_Symbol] :=
  Log[a+b*ArcSinh[c*x]]/(b*c*Sqrt[d]) /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && GtQ[d,0]
```

```
Int[1/(Sqrt[d1_+e1_.*x_]*Sqrt[d2_+e2_.*x_]*(a_.+b_.*ArcCosh[c_.*x_])),x_Symbol] :=
  Log[a+b*ArcCosh[c*x]]/(b*c*Sqrt[-d1*d2]) /;
FreeQ[{a,b,c,d1,e1,d2,e2},x] && EqQ[e1,c*d1] && EqQ[e2,-c*d2] && GtQ[d1,0] && LtQ[d2,0]
```

2: $\int \frac{(a+b\,\text{ArcSinh}[c\,x])^n}{\sqrt{d+e x^2}}\,dx$ when $e == c^2 d \wedge d > 0 \wedge n \neq -1$

- **Derivation: Integration by substitution**
- **Rule: If $e == c^2 d \wedge d > 0 \wedge n \neq -1$, then**

$$\int \frac{(a+b\,\text{ArcSinh}[c\,x])^n}{\sqrt{d+e x^2}}\,dx \longrightarrow \frac{(a+b\,\text{ArcSinh}[c\,x])^{n+1}}{b\,c\,\sqrt{d}\,(n+1)}$$

- **Program code:**

```
Int[(a_.+b_.*ArcSinh[c_.*x_])^n_./Sqrt[d_+e_.*x_^2],x_Symbol] :=
  (a+b*ArcSinh[c*x])^(n+1)/(b*c*Sqrt[d]*(n+1)) /;
FreeQ[{a,b,c,d,e,n},x] && EqQ[e,c^2*d] && GtQ[d,0] && NeQ[n,-1]
```

```
Int[(a_.+b_.*ArcCosh[c_.*x_])^n_./(Sqrt[d1_+e1_.*x_]*Sqrt[d2_+e2_.*x_]),x_Symbol] :=
  (a+b*ArcCosh[c*x])^(n+1)/(b*c*Sqrt[-d1*d2]*(n+1)) /;
FreeQ[{a,b,c,d1,e1,d2,e2,n},x] && EqQ[e1,c*d1] && EqQ[e2,-c*d2] && GtQ[d1,0] && LtQ[d2,0] && NeQ[n,-1]
```

2: $\int \frac{(a + b \,\text{ArcSinh}[c\,x])^n}{\sqrt{d + e\,x^2}}\,dx$ when $e == c^2 d \wedge d \ngtr 0$

Derivation: Piecewise constant extraction

- **Basis: If** $e == c^2 d$**, then** $\partial_x \frac{\sqrt{1+c^2 x^2}}{\sqrt{d+e\,x^2}} == 0$

Rule: If $e == c^2 d \wedge d \ngtr 0$**, then**

$$\int \frac{(a + b \,\text{ArcSinh}[c\,x])^n}{\sqrt{d + e\,x^2}}\,dx \longrightarrow \frac{\sqrt{1 + c^2 x^2}}{\sqrt{d + e\,x^2}} \int \frac{(a + b \,\text{ArcSinh}[c\,x])^n}{\sqrt{1 + c^2 x^2}}\,dx$$

Program code:

```
Int[(a_.+b_.*ArcSinh[c_.*x_])^n_./Sqrt[d_+e_.*x_^2],x_Symbol] :=
  Sqrt[1+c^2*x^2]/Sqrt[d+e*x^2]*Int[(a+b*ArcSinh[c*x])^n/Sqrt[1+c^2*x^2],x] /;
FreeQ[{a,b,c,d,e,n},x] && EqQ[e,c^2*d] && Not[GtQ[d,0]]
```

```
Int[(a_.+b_.*ArcCosh[c_.*x_])^n_./(Sqrt[d1_+e1_.*x_]*Sqrt[d2_+e2_.*x_]),x_Symbol] :=
  Sqrt[1+c*x]*Sqrt[-1+c*x]/(Sqrt[d1+e1*x]*Sqrt[d2+e2*x])*Int[(a+b*ArcCosh[c*x])^n/(Sqrt[1+c*x]*Sqrt[-1+c*x]),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,n},x] && EqQ[e1,c*d1] && EqQ[e2,-c*d2] && Not[GtQ[d1,0] && LtQ[d2,0]]
```

2. $\int (d + e\,x^2)^p\,(a + b \,\text{ArcSinh}[c\,x])^n\,dx$ when $e == c^2 d \wedge n > 0$

1: $\int (d + e\,x^2)^p\,(a + b \,\text{ArcSinh}[c\,x])\,dx$ when $e == c^2 d \wedge p \in \mathbb{Z}^+$

Derivation: Integration by parts

Rule: If $e == c^2 d \wedge p \in \mathbb{Z}^+$**, let** $u = \int (d + e\,x^2)^p\,dx$**, then**

$$\int (d + e\,x^2)^p\,(a + b \,\text{ArcSinh}[c\,x])\,dx \longrightarrow u\,(a + b \,\text{ArcSinh}[c\,x]) - b\,c \int \frac{u}{\sqrt{1 + c^2 x^2}}\,dx$$

Program code:

```
Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[(d+e*x^2)^p,x]},
  Dist[a+b*ArcSinh[c*x],u,x] - b*c*Int[SimplifyIntegrand[u/Sqrt[1+c^2*x^2],x],x]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && IGtQ[p,0]
```

```
Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCosh[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[(d+e*x^2)^p,x]},
  Dist[a+b*ArcCosh[c*x],u,x] - b*c*Int[SimplifyIntegrand[u/(Sqrt[1+c*x]*Sqrt[-1+c*x]),x],x]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[c^2*d+e,0] && IGtQ[p,0]
```

2. $\int (d+e x^2)^p (a+b \text{ArcSinh}[c x])^n \, dx$ when $e == c^2 d \land n > 0 \land p > 0$

1: $\int (d+e x^2)^p (a+b \text{ArcSinh}[c x])^n \, dx$ when $e == c^2 d \land n > 0 \land p > 0 \land (p \in \mathbb{Z} \lor d > 0)$

Derivation: Inverted integration by parts

Rule: If $e == c^2 d \land n > 0 \land p > 0 \land (p \in \mathbb{Z} \lor d > 0)$**, then**

$$\int (d+e x^2)^p (a+b \text{ArcSinh}[c x])^n \, dx \longrightarrow$$
$$\frac{x (d+e x^2)^p (a+b \text{ArcSinh}[c x])^n}{2 p+1} +$$
$$\frac{2 d p}{2 p+1} \int (d+e x^2)^{p-1} (a+b \text{ArcSinh}[c x])^n \, dx - \frac{b c n d^p}{2 p+1} \int x (1+c^2 x^2)^{p-\frac{1}{2}} (a+b \text{ArcSinh}[c x])^{n-1} \, dx$$

Program code:

```
(* Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  x*(d+e*x^2)^p*(a+b*ArcSinh[c*x])^n/(2*p+1) +
  2*d*p/(2*p+1)*Int[(d+e*x^2)^(p-1)*(a+b*ArcSinh[c*x])^n,x] -
  b*c*n*d^p/(2*p+1)*Int[x*(1+c^2*x^2)^(p-1/2)*(a+b*ArcSinh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && GtQ[n,0] && GtQ[p,0] && (IntegerQ[p] || GtQ[d,0]) *)
```

```
Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  x*(d+e*x^2)^p*(a+b*ArcCosh[c*x])^n/(2*p+1) +
  2*d*p/(2*p+1)*Int[(d+e*x^2)^(p-1)*(a+b*ArcCosh[c*x])^n,x] -
  b*c*n*(-d)^p/((2*p+1))*Int[x*(-1+c*x)^(p-1/2)*(1+c*x)^(p-1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[c^2*d+e,0] && GtQ[n,0] && GtQ[p,0] && IntegerQ[p]
```

```
(* Int[(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  x*(d1+e1*x)^p*(d2+e2*x)^p*(a+b*ArcCosh[c*x])^n/(2*p+1) +
  2*d1*d2*p/(2*p+1)*Int[(d1+e1*x)^(p-1)*(d2+e2*x)^(p-1)*(a+b*ArcCosh[c*x])^n,x] -
  b*c*n*(-d1*d2)^p/((2*p+1))*Int[x*(-1+c^2*x^2)^(p-1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2},x] && EqQ[e1,c*d1] && EqQ[e2,-c*d2] && GtQ[n,0] && GtQ[p,0] && IntegerQ[p-1/2] && (GtQ[d1,0] && LtQ[d2,0]
```

2. $\int (d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e == c^2\,d\,\wedge\,n>0\,\wedge\,p>0$

1: $\int \sqrt{d+e\,x^2}\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e == c^2\,d\,\wedge\,n>0$

Derivation: Inverted integration by parts

Note: The piecewise constant factor in the second integral reduces the degree of d in the resulting antiderivative.

Rule: If $e == c^2\,d\,\wedge\,n>0$, then

$$\int \sqrt{d+e\,x^2}\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx\,\longrightarrow$$

$$\frac{x\sqrt{d+e\,x^2}\,(a+b\,\text{ArcSinh}[c\,x])^n}{2}-\frac{b\,c\,n\sqrt{d+e\,x^2}}{2\sqrt{1+c^2\,x^2}}\int x\,(a+b\,\text{ArcSinh}[c\,x])^{n-1}\,dx+\frac{\sqrt{d+e\,x^2}}{2\sqrt{1+c^2\,x^2}}\int\frac{(a+b\,\text{ArcSinh}[c\,x])^n}{\sqrt{1+c^2\,x^2}}\,dx$$

Program code:

```
Int[Sqrt[d_+e_.*x_^2]*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  x*Sqrt[d+e*x^2]*(a+b*ArcSinh[c*x])^n/2 -
  b*c*n*Sqrt[d+e*x^2]/(2*Sqrt[1+c^2*x^2])*Int[x*(a+b*ArcSinh[c*x])^(n-1),x] +
  Sqrt[d+e*x^2]/(2*Sqrt[1+c^2*x^2])*Int[(a+b*ArcSinh[c*x])^n/Sqrt[1+c^2*x^2],x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && GtQ[n,0]
```

```
Int[Sqrt[d1_+e1_.*x_]*Sqrt[d2_+e2_.*x_]*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  x*Sqrt[d1+e1*x]*Sqrt[d2+e2*x]*(a+b*ArcCosh[c*x])^n/2 -
  b*c*n*Sqrt[d1+e1*x]*Sqrt[d2+e2*x]/(2*Sqrt[1+c*x]*Sqrt[-1+c*x])*Int[x*(a+b*ArcCosh[c*x])^(n-1),x] -
  Sqrt[d1+e1*x]*Sqrt[d2+e2*x]/(2*Sqrt[1+c*x]*Sqrt[-1+c*x])*Int[(a+b*ArcCosh[c*x])^n/(Sqrt[1+c*x]*Sqrt[-1+c*x]),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2},x] && EqQ[e1,c*d1] && EqQ[e2,-c*d2] && GtQ[n,0]
```

2: $\int (d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e == c^2\,d\,\wedge\,n>0\,\wedge\,p>0$

Derivation: Inverted integration by parts and piecewise constant extraction

Basis: If $e == c^2\,d$, then $\partial_x\frac{(d+e\,x^2)^p}{(1+c^2\,x^2)^p} == 0$

Rule: If $e == c^2\,d\,\wedge\,n>0\,\wedge\,p>0$, then

$$\int (d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx\,\longrightarrow$$

$$\frac{x\left(d+e x^2\right)^p (a+b\,\mathrm{ArcSinh}[c\,x])^n}{2p+1}+$$

$$\frac{2dp}{2p+1}\int\left(d+e x^2\right)^{p-1}(a+b\,\mathrm{ArcSinh}[c\,x])^n\,dx-\frac{b\,c\,n\,d^{\mathrm{IntPart}[p]}\left(d+e x^2\right)^{\mathrm{FracPart}[p]}}{(2p+1)\left(1+c^2x^2\right)^{\mathrm{FracPart}[p]}}\int x\left(1+c^2x^2\right)^{p-\frac{1}{2}}(a+b\,\mathrm{ArcSinh}[c\,x])^{n-1}\,dx$$

Program code:

```
Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  x*(d+e*x^2)^p*(a+b*ArcSinh[c*x])^n/(2*p+1) +
  2*d*p/(2*p+1)*Int[(d+e*x^2)^(p-1)*(a+b*ArcSinh[c*x])^n,x] -
   b*c*n*d^IntPart[p]*(d+e*x^2)^FracPart[p]/((2*p+1)*(1+c^2*x^2)^FracPart[p])*
    Int[x*(1+c^2*x^2)^(p-1/2)*(a+b*ArcSinh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && GtQ[n,0] && GtQ[p,0]
```

```
Int[(d1_+e1_.*x_)^p_.*(d2_+e2_.*x_)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  x*(d1+e1*x)^p*(d2+e2*x)^p*(a+b*ArcCosh[c*x])^n/(2*p+1) +
  2*d1*d2*p/(2*p+1)*Int[(d1+e1*x)^(p-1)*(d2+e2*x)^(p-1)*(a+b*ArcCosh[c*x])^n,x] -
  b*c*n*(-d1*d2)^(p-1/2)*Sqrt[d1+e1*x]*Sqrt[d2+e2*x]/((2*p+1)*Sqrt[1+c*x]*Sqrt[-1+c*x])*
    Int[x*(-1+c^2*x^2)^(p-1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2},x] && EqQ[e1,c*d1] && EqQ[e2,-c*d2] && GtQ[n,0] && GtQ[p,0] && IntegerQ[p-1/2]
```

```
Int[(d1_+e1_.*x_)^p_.*(d2_+e2_.*x_)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  x*(d1+e1*x)^p*(d2+e2*x)^p*(a+b*ArcCosh[c*x])^n/(2*p+1) +
  2*d1*d2*p/(2*p+1)*Int[(d1+e1*x)^(p-1)*(d2+e2*x)^(p-1)*(a+b*ArcCosh[c*x])^n,x] -
  b*c*n*(-d1*d2)^IntPart[p]*(d1+e1*x)^FracPart[p]*(d2+e2*x)^FracPart[p]/((2*p+1)*(1+c*x)^FracPart[p]*(-1+c*x)^FracPart[p])*
    Int[x*(-1+c*x)^(p-1/2)*(1+c*x)^(p-1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2},x] && EqQ[e1,c*d1] && EqQ[e2,-c*d2] && GtQ[n,0] && GtQ[p,0]
```

3. $\int (d+e x^2)^p (a+b \text{ArcSinh}[c x])^n \, dx$ when $e == c^2 d \land n > 0 \land p < -1$

1. $\int \frac{(a+b \text{ArcSinh}[c x])^n}{(d+e x^2)^{3/2}} \, dx$ when $e == c^2 d \land n > 0$

1: $\int \frac{(a+b \text{ArcSinh}[c x])^n}{(d+e x^2)^{3/2}} \, dx$ when $e == c^2 d \land n > 0 \land d > 0$

Derivation: Integration by parts

Basis: $\frac{1}{(d+e x^2)^{3/2}} == \partial_x \frac{x}{d \sqrt{d+e x^2}}$

Rule: If $e == c^2 d \land n > 0 \land d > 0$**, then**

$$\int \frac{(a+b \text{ArcSinh}[c x])^n}{(d+e x^2)^{3/2}} \, dx \to \frac{x (a+b \text{ArcSinh}[c x])^n}{d \sqrt{d+e x^2}} - \frac{b c n}{\sqrt{d}} \int \frac{x (a+b \text{ArcSinh}[c x])^{n-1}}{d+e x^2} \, dx$$

Program code:

```
(* Int[(a_.+b_.*ArcSinh[c_.*x_])^n_./(d_+e_.*x_^2)^(3/2),x_Symbol] :=
  x*(a+b*ArcSinh[c*x])^n/(d*Sqrt[d+e*x^2]) -
  b*c*n/Sqrt[d]*Int[x*(a+b*ArcSinh[c*x])^(n-1)/(d+e*x^2),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && GtQ[n,0] && GtQ[d,0] *)
```

```
(* Int[(a_.+b_.*ArcCosh[c_.*x_])^n_./((d1_+e1_.*x_)^(3/2)*(d2_+e2_.*x_)^(3/2)),x_Symbol] :=
  x*(a+b*ArcCosh[c*x])^n/(d1*d2*Sqrt[d1+e1*x]*Sqrt[d2+e2*x]) +
  b*c*n/Sqrt[-d1*d2]*Int[x*(a+b*ArcCosh[c*x])^(n-1)/(d1*d2+e1*e2*x^2),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2},x] && EqQ[e1,c*d1] && EqQ[e2,-c*d2] && GtQ[n,0] && (GtQ[d1,0] && LtQ[d2,0]) *)
```

2: $\int \frac{(a+b \text{ArcSinh}[c x])^n}{(d+e x^2)^{3/2}} \, dx$ when $e == c^2 d \land n > 0$

Derivation: Integration by parts and piecewise constant extraction

Basis: $\frac{1}{(d+e x^2)^{3/2}} == \partial_x \frac{x}{d \sqrt{d+e x^2}}$

Basis: If $e == c^2 d$**, then** $\partial_x \frac{\sqrt{1+c^2 x^2}}{\sqrt{d+e x^2}} == 0$

Rule: If $e == c^2 d \land n > 0$**, then**

$$\int \frac{(a+b\,\text{ArcSinh}[c\,x])^n}{\left(d+e\,x^2\right)^{3/2}}\,dx\;\longrightarrow\;\frac{x\,(a+b\,\text{ArcSinh}[c\,x])^n}{d\,\sqrt{d+e\,x^2}}-\frac{b\,c\,n\,\sqrt{1+c^2\,x^2}}{d\,\sqrt{d+e\,x^2}}\int \frac{x\,(a+b\,\text{ArcSinh}[c\,x])^{n-1}}{1+c^2\,x^2}\,dx$$

Program code:

```
Int[(a_.+b_.*ArcSinh[c_.*x_])^n_./(d_+e_.*x_^2)^(3/2),x_Symbol] :=
  x*(a+b*ArcSinh[c*x])^n/(d*Sqrt[d+e*x^2]) -
  b*c*n*Sqrt[1+c^2*x^2]/(d*Sqrt[d+e*x^2])*Int[x*(a+b*ArcSinh[c*x])^(n-1)/(1+c^2*x^2),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && GtQ[n,0]
```

```
Int[(a_.+b_.*ArcCosh[c_.*x_])^n_./((d1_+e1_.*x_)^(3/2)*(d2_+e2_.*x_)^(3/2)),x_Symbol] :=
  x*(a+b*ArcCosh[c*x])^n/(d1*d2*Sqrt[d1+e1*x]*Sqrt[d2+e2*x]) +
  b*c*n*Sqrt[1+c*x]*Sqrt[-1+c*x]/(d1*d2*Sqrt[d1+e1*x]*Sqrt[d2+e2*x])*Int[x*(a+b*ArcCosh[c*x])^(n-1)/(1-c^2*x^2),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2},x] && EqQ[e1,c*d1] && EqQ[e2,-c*d2] && GtQ[n,0]
```

2. $\int \left(d+e\,x^2\right)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e == c^2\,d\;\bigwedge\;n>0\;\bigwedge\;p<-1\;\bigwedge\;p\neq-\frac{3}{2}$

1: $\int \left(d+e\,x^2\right)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e == c^2\,d\;\bigwedge\;n>0\;\bigwedge\;p<-1\;\bigwedge\;p\neq-\frac{3}{2}\;\bigwedge\;(p\in\mathbb{Z}\;\bigvee\;d>0)$

-

Rule: If $e == c^2\,d\;\bigwedge\;n>0\;\bigwedge\;p<-1\;\bigwedge\;p\neq-\frac{3}{2}\;\bigwedge\;(p\in\mathbb{Z}\;\bigvee\;d>0)$, **then**

$$\int \left(d+e\,x^2\right)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx\;\longrightarrow$$

$$-\frac{x\,\left(d+e\,x^2\right)^{p+1}\,(a+b\,\text{ArcSinh}[c\,x])^n}{2\,d\,(p+1)}+$$

$$\frac{2\,p+3}{2\,d\,(p+1)}\int \left(d+e\,x^2\right)^{p+1}\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx+\frac{b\,c\,n\,d^p}{2\,(p+1)}\int x\,\left(1+c^2\,x^2\right)^{p+\frac{1}{2}}\,(a+b\,\text{ArcSinh}[c\,x])^{n-1}\,dx$$

Program code:

```
(* Int[(d_+e_.*x_^2)^p_*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  -x*(d+e*x^2)^(p+1)*(a+b*ArcSinh[c*x])^n/(2*d*(p+1)) +
  (2*p+3)/(2*d*(p+1))*Int[(d+e*x^2)^(p+1)*(a+b*ArcSinh[c*x])^n,x] +
  b*c*n*d^p/(2*(p+1))*Int[x*(1+c^2*x^2)^(p+1/2)*(a+b*ArcSinh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && GtQ[n,0] && LtQ[p,-1] && NeQ[p,-3/2] && (IntegerQ[p] || GtQ[d,0]) *)
```

```
Int[(d_+e_.*x_^2)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  -x*(d+e*x^2)^(p+1)*(a+b*ArcCosh[c*x])^n/(2*d*(p+1)) +
  (2*p+3)/(2*d*(p+1))*Int[(d+e*x^2)^(p+1)*(a+b*ArcCosh[c*x])^n,x] -
  b*c*n*(-d)^p/(2*(p+1))*Int[x*(1+c*x)^(p+1/2)*(-1+c*x)^(p+1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[c^2*d+e,0] && GtQ[n,0] && LtQ[p,-1] && IntegerQ[p]
```

```
(* Int[(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  -x*(d1+e1*x)^(p+1)*(d2+e2*x)^(p+1)*(a+b*ArcCosh[c*x])^n/(2*d1*d2*(p+1)) +
  (2*p+3)/(2*d1*d2*(p+1))*Int[(d1+e1*x)^(p+1)*(d2+e2*x)^(p+1)*(a+b*ArcCosh[c*x])^n,x] -
  b*c*n*(-d1*d2)^p/(2*(p+1))*Int[x*(-1+c^2*x^2)^(p+1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2},x] && EqQ[e1,c*d1] && EqQ[e2,-c*d2] && GtQ[n,0] && LtQ[p,-1] && NeQ[p,-3/2] &&
  IntegerQ[p+1/2] && (GtQ[d1,0] && LtQ[d2,0]) *)
```

2: $\int \left(d+e\,x^2\right)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e==c^2\,d\ \bigwedge\ n>0\ \bigwedge\ p<-1\ \bigwedge\ p\neq-\frac{3}{2}$

- **Rule: If** $e==c^2\,d\ \bigwedge\ n>0\ \bigwedge\ p<-1\ \bigwedge\ p\neq-\frac{3}{2}$**, then**

$$\int \left(d+e\,x^2\right)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx\ \longrightarrow$$

$$-\frac{x\left(d+e\,x^2\right)^{p+1}\,(a+b\,\text{ArcSinh}[c\,x])^n}{2\,d\,(p+1)}+\frac{2\,p+3}{2\,d\,(p+1)}\int\left(d+e\,x^2\right)^{p+1}\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx+$$
$$\frac{b\,c\,n\,d^{\text{IntPart}[p]}\,\left(d+e\,x^2\right)^{\text{FracPart}[p]}}{2\,(p+1)\,\left(1+c^2\,x^2\right)^{\text{FracPart}[p]}}\int x\left(1+c^2\,x^2\right)^{p+\frac{1}{2}}\,(a+b\,\text{ArcSinh}[c\,x])^{n-1}\,dx$$

Program code:

```
Int[(d_+e_.*x_^2)^p_*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  -x*(d+e*x^2)^(p+1)*(a+b*ArcSinh[c*x])^n/(2*d*(p+1)) +
  (2*p+3)/(2*d*(p+1))*Int[(d+e*x^2)^(p+1)*(a+b*ArcSinh[c*x])^n,x] +
  b*c*n*d^IntPart[p]*(d+e*x^2)^FracPart[p]/(2*(p+1)*(1+c^2*x^2)^FracPart[p])*
    Int[x*(1+c^2*x^2)^(p+1/2)*(a+b*ArcSinh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && GtQ[n,0] && LtQ[p,-1] && NeQ[p,-3/2]
```

```
Int[(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  -x*(d1+e1*x)^(p+1)*(d2+e2*x)^(p+1)*(a+b*ArcCosh[c*x])^n/(2*d1*d2*(p+1)) +
  (2*p+3)/(2*d1*d2*(p+1))*Int[(d1+e1*x)^(p+1)*(d2+e2*x)^(p+1)*(a+b*ArcCosh[c*x])^n,x] -
  b*c*n*(-d1*d2)^(p+1/2)*Sqrt[1+c*x]*Sqrt[-1+c*x]/(2*(p+1)*Sqrt[d1+e1*x]*Sqrt[d2+e2*x])*
    Int[x*(-1+c^2*x^2)^(p+1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2},x] && EqQ[e1,c*d1] && EqQ[e2,-c*d2] && GtQ[n,0] && LtQ[p,-1] && NeQ[p,-3/2] && IntegerQ[p+1/2]
```

```
Int[(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  -x*(d1+e1*x)^(p+1)*(d2+e2*x)^(p+1)*(a+b*ArcCosh[c*x])^n/(2*d1*d2*(p+1)) +
  (2*p+3)/(2*d1*d2*(p+1))*Int[(d1+e1*x)^(p+1)*(d2+e2*x)^(p+1)*(a+b*ArcCosh[c*x])^n,x] -
  b*c*n*(-d1*d2)^IntPart[p]*(d1+e1*x)^FracPart[p]*(d2+e2*x)^FracPart[p]/(2*(p+1)*(1+c*x)^FracPart[p]*(-1+c*x)^FracPart[p])*
    Int[x*(1+c*x)^(p+1/2)*(-1+c*x)^(p+1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2},x] && EqQ[e1,c*d1] && EqQ[e2,-c*d2] && GtQ[n,0] && LtQ[p,-1] && NeQ[p,-3/2]
```

4: $\int \frac{(a+b\,\text{ArcSinh}[c\,x])^n}{d+e\,x^2}\,dx$ when $e == c^2 d \land n \in \mathbb{Z}^+$

Derivation: Integration by substitution

Basis: If $e == c^2 d$**, then** $\frac{1}{d+e\,x^2} == \frac{1}{c\,d}\,\text{Subst}[\text{Sech}[x],\,x,\,\text{ArcSinh}[c\,x]]\,\partial_x \text{ArcSinh}[c\,x]$

Note: If $n \in \mathbb{Z}^+$**, then** $(a+b\,x)^n\,\text{Sech}[x]$ **is integrable in closed-form.**

Rule: If $e == c^2 d \land n \in \mathbb{Z}^+$**, then**

$$\int \frac{(a+b\,\text{ArcSinh}[c\,x])^n}{d+e\,x^2}\,dx \longrightarrow \frac{1}{c\,d}\,\text{Subst}\left[\int (a+b\,x)^n\,\text{Sech}[x]\,dx,\,x,\,\text{ArcSinh}[c\,x]\right]$$

Program code:

```
Int[(a_.+b_.*ArcSinh[c_.*x_])^n_./(d_+e_.*x_^2),x_Symbol] :=
  1/(c*d)*Subst[Int[(a+b*x)^n*Sech[x],x],x,ArcSinh[c*x]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && IGtQ[n,0]
```

```
Int[(a_.+b_.*ArcCosh[c_.*x_])^n_./(d_+e_.*x_^2),x_Symbol] :=
  -1/(c*d)*Subst[Int[(a+b*x)^n*Csch[x],x],x,ArcCosh[c*x]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[c^2*d+e,0] && IGtQ[n,0]
```

3. $\int (d + e x^2)^p (a + b \text{ArcSinh}[c x])^n \, dx$ when $e == c^2 d \land n < -1$

1: $\int (d + e x^2)^p (a + b \text{ArcSinh}[c x])^n \, dx$ when $e == c^2 d \land n < -1 \land (p \in \mathbb{Z} \lor d > 0)$

Derivation: Integration by parts

- **Basis:** $\frac{(a+b \text{ArcSinh}[c x])^n}{\sqrt{1+c^2 x^2}} == \partial_x \frac{(a+b \text{ArcSinh}[c x])^{n+1}}{b c (n+1)}$

- **Rule: If** $e == c^2 d \land n < -1 \land (p \in \mathbb{Z} \lor d > 0)$**, then**

$$\int (d + e x^2)^p (a + b \text{ArcSinh}[c x])^n \, dx \longrightarrow$$

$$\frac{d^p \left(1 + c^2 x^2\right)^{p+\frac{1}{2}} (a + b \text{ArcSinh}[c x])^{n+1}}{b c (n + 1)} - \frac{c d^p (2 p + 1)}{b (n + 1)} \int x \left(1 + c^2 x^2\right)^{p-\frac{1}{2}} (a + b \text{ArcSinh}[c x])^{n+1} \, dx$$

- **Program code:**

```
(* Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_])^n_,x_Symbol] :=
  d^p*(1+c^2*x^2)^(p+1/2)*(a+b*ArcSinh[c*x])^(n+1)/(b*c*(n+1)) -
  c*d^p*(2*p+1)/(b*(n+1))*Int[x*(1+c^2*x^2)^(p-1/2)*(a+b*ArcSinh[c*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,p},x] && EqQ[e,c^2*d] && LtQ[n,-1] && (IntegerQ[p] || GtQ[d,0]) *)
```

```
Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_,x_Symbol] :=
  (-d)^p*(-1+c*x)^(p+1/2)*(1+c*x)^(p+1/2)*(a+b*ArcCosh[c*x])^(n+1)/(b*c*(n+1)) -
  c*(-d)^p*(2*p+1)/(b*(n+1))*Int[x*(-1+c*x)^(p-1/2)*(1+c*x)^(p-1/2)*(a+b*ArcCosh[c*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,p},x] && EqQ[c^2*d+e,0] && LtQ[n,-1] && IntegerQ[p]
```

```
(* Int[(d1_+e1_.*x_)^p_.*(d2_+e2_.*x_)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_,x_Symbol] :=
  (-d1*d2)^p*(-1+c*x)^(p+1/2)*(1+c*x)^(p+1/2)*(a+b*ArcCosh[c*x])^(n+1)/(b*c*(n+1)) -
  c*(-d1*d2)^p*(2*p+1)/(b*(n+1))*Int[x*(-1+c^2*x^2)^(p-1/2)*(a+b*ArcCosh[c*x])^(n+1),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,p},x] && EqQ[e1,c*d1] && EqQ[e2,-c*d2] && LtQ[n,-1] && IntegerQ[p-1/2] && (GtQ[d1,0] && LtQ[d2,0]) *)
```

2: $\int \left(d+e\,x^2\right)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e == c^2\,d \,\wedge\, n<-1$

Derivation: Integration by parts and piecewise constant extraction

- **Basis:** $\frac{(a+b\,\text{ArcSinh}[c\,x])^n}{\sqrt{1+c^2\,x^2}} == \partial_x \frac{(a+b\,\text{ArcSinh}[c\,x])^{n+1}}{b\,c\,(n+1)}$
- **Basis: If** $e == c^2\,d$, **then** $\partial_x \frac{(d+e\,x^2)^p}{(1+c^2\,x^2)^p} == 0$
- **Rule: If** $e == c^2\,d \,\wedge\, n<-1$, **then**

$$\int \left(d+e\,x^2\right)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx \longrightarrow$$

$$\frac{\sqrt{1+c^2\,x^2}\,\left(d+e\,x^2\right)^p\,(a+b\,\text{ArcSinh}[c\,x])^{n+1}}{b\,c\,(n+1)}+\frac{c\,(2\,p+1)}{b\,(n+1)}\int \frac{x\,\left(d+e\,x^2\right)^p\,(a+b\,\text{ArcSinh}[c\,x])^{n+1}}{\sqrt{1+c^2\,x^2}}\,dx \longrightarrow$$

$$\frac{\sqrt{1+c^2\,x^2}\,\left(d+e\,x^2\right)^p\,(a+b\,\text{ArcSinh}[c\,x])^{n+1}}{b\,c\,(n+1)}-\frac{c\,(2\,p+1)\,d^{\text{IntPart}[p]}\,\left(d+e\,x^2\right)^{\text{FracPart}[p]}}{b\,(n+1)\,\left(1+c^2\,x^2\right)^{\text{FracPart}[p]}}\int x\,\left(1+c^2\,x^2\right)^{p-\frac{1}{2}}\,(a+b\,\text{ArcSinh}[c\,x])^{n+1}\,dx$$

Program code:

```
Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_])^n_,x_Symbol] :=
  Sqrt[1+c^2*x^2]*(d+e*x^2)^p*(a+b*ArcSinh[c*x])^(n+1)/(b*c*(n+1)) -
  c*(2*p+1)*d^IntPart[p]*(d+e*x^2)^FracPart[p]/(b*(n+1)*(1+c^2*x^2)^FracPart[p])*
    Int[x*(1+c^2*x^2)^(p-1/2)*(a+b*ArcSinh[c*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,p},x] && EqQ[e,c^2*d] && LtQ[n,-1]
```

```
Int[(d1_+e1_.*x_)^p_.*(d2_+e2_.*x_)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_,x_Symbol] :=
  Sqrt[1+c*x]*Sqrt[-1+c*x]*(d1+e1*x)^p*(d2+e2*x)^p*(a+b*ArcCosh[c*x])^(n+1)/(b*c*(n+1)) -
  c*(2*p+1)*(-d1*d2)^(p-1/2)*Sqrt[d1+e1*x]*Sqrt[d2+e2*x]/(b*(n+1)*Sqrt[1+c*x]*Sqrt[-1+c*x])*
    Int[x*(-1+c^2*x^2)^(p-1/2)*(a+b*ArcCosh[c*x])^(n+1),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,p},x] && EqQ[e1,c*d1] && EqQ[e2,-c*d2] && LtQ[n,-1] && IntegerQ[p-1/2]
```

```
Int[(d1_+e1_.*x_)^p_.*(d2_+e2_.*x_)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_,x_Symbol] :=
  Sqrt[1+c*x]*Sqrt[-1+c*x]*(d1+e1*x)^p*(d2+e2*x)^p*(a+b*ArcCosh[c*x])^(n+1)/(b*c*(n+1)) -
  c*(2*p+1)*(-d1*d2)^IntPart[p]*(d1+e1*x)^FracPart[p]*(d2+e2*x)^FracPart[p]/(b*(n+1)*(1+c*x)^FracPart[p]*(-1+c*x)^FracPart[p])*
    Int[x*(1+c*x)^(p-1/2)*(-1+c*x)^(p-1/2)*(a+b*ArcCosh[c*x])^(n+1),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,p},x] && EqQ[e1,c*d1] && EqQ[e2,-c*d2] && LtQ[n,-1]
```

4. $\int (d+e x^2)^p (a+b \text{ArcSinh}[c x])^n \, dx$ when $e == c^2 d \wedge 2p \in \mathbb{Z}^+$

1: $\int (d+e x^2)^p (a+b \text{ArcSinh}[c x])^n \, dx$ when $e == c^2 d \wedge 2p \in \mathbb{Z}^+ \wedge (p \in \mathbb{Z} \vee d > 0)$

Derivation: Integration by substitution

Basis: If $e == c^2 d \wedge (p \in \mathbb{Z} \vee d > 0)$**, then** $(d+e x^2)^p == \frac{d^p}{c} \text{Subst}\left[\text{Cosh}[x]^{2p+1}, x, \text{ArcSinh}[c x]\right] \partial_x \text{ArcSinh}[c x]$

Note: If $2p \in \mathbb{Z}^+$**, then** $(a+b x)^n \text{Cosh}[x]^{2p+1}$ **is integrable in closed-form.**

Rule: If $e == c^2 d \wedge 2p \in \mathbb{Z}^+ \wedge (p \in \mathbb{Z} \vee d > 0)$**, then**

$$\int (d+e x^2)^p (a+b \text{ArcSinh}[c x])^n \, dx \longrightarrow \frac{d^p}{c} \text{Subst}\left[\int (a+b x)^n \text{Cosh}[x]^{2p+1} \, dx, x, \text{ArcSinh}[c x]\right]$$

Program code:

```
Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  d^p/c*Subst[Int[(a+b*x)^n*Cosh[x]^(2*p+1),x],x,ArcSinh[c*x]] /;
FreeQ[{a,b,c,d,e,n},x] && EqQ[e,c^2*d] && IGtQ[2*p,0] && (IntegerQ[p] || GtQ[d,0])
```

```
Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (-d)^p/c*Subst[Int[(a+b*x)^n*Sinh[x]^(2*p+1),x],x,ArcCosh[c*x]] /;
FreeQ[{a,b,c,d,e,n},x] && EqQ[c^2*d+e,0] && IGtQ[p,0]
```

```
Int[(d1_+e1_.*x_)^p_.*(d2_+e2_.*x_)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (-d1*d2)^p/c*Subst[Int[(a+b*x)^n*Sinh[x]^(2*p+1),x],x,ArcCosh[c*x]] /;
FreeQ[{a,b,c,d1,e1,d2,e2,n},x] && EqQ[e1,c*d1] && EqQ[e2,-c*d2] && IGtQ[p+1/2,0] && (GtQ[d1,0] && LtQ[d2,0])
```

2: $\int (d+e x^2)^p (a+b \operatorname{ArcSinh}[c x])^n \, dx$ when $e == c^2 d \wedge 2 p \in \mathbb{Z}^+ \wedge \neg (p \in \mathbb{Z} \vee d > 0)$

- **Derivation: Piecewise constant extraction**
- **Basis: If** $e == c^2 d$, **then** $\partial_x \frac{\sqrt{d+e x^2}}{\sqrt{1+c^2 x^2}} == 0$
- **Rule: If** $e == c^2 d \wedge 2 p \in \mathbb{Z}^+ \wedge \neg (p \in \mathbb{Z} \vee d > 0)$**, then**

$$\int (d+e x^2)^p (a+b \operatorname{ArcSinh}[c x])^n \, dx \longrightarrow \frac{d^{p-\frac{1}{2}} \sqrt{d+e x^2}}{\sqrt{1+c^2 x^2}} \int (1+c^2 x^2)^p (a+b \operatorname{ArcSinh}[c x])^n \, dx$$

- **Program code:**

```
Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  d^(p-1/2)*Sqrt[d+e*x^2]/Sqrt[1+c^2*x^2]*Int[(1+c^2*x^2)^p*(a+b*ArcSinh[c*x])^n,x] /;
FreeQ[{a,b,c,d,e,n},x] && EqQ[e,c^2*d] && IGtQ[2*p,0] && Not[IntegerQ[p] || GtQ[d,0]]
```

```
Int[(d1_+e1_.*x_)^p_.*(d2_+e2_.*x_)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (-d1*d2)^(p-1/2)*Sqrt[d1+e1*x]*Sqrt[d2+e2*x]/(Sqrt[1+c*x]*Sqrt[-1+c*x])*Int[(1+c*x)^p*(-1+c*x)^p*(a+b*ArcCosh[c*x])^n,x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,n},x] && EqQ[e1,c*d1] && EqQ[e2,-c*d2] && IGtQ[2*p,0] && Not[GtQ[d1,0] && LtQ[d2,0]]
```

2. $\int (d+e x^2)^p (a+b \operatorname{ArcSinh}[c x])^n \, dx$ when $e \neq c^2 d$

1: $\int (d+e x^2)^p (a+b \operatorname{ArcSinh}[c x]) \, dx$ when $e \neq c^2 d \wedge \left(p \in \mathbb{Z}^+ \vee p+\frac{1}{2} \in \mathbb{Z}^-\right)$

Derivation: Integration by parts

- **Note: If** $p \in \mathbb{Z}^+ \vee p+\frac{1}{2} \in \mathbb{Z}^-$**, then** $\int (d+e x^2)^p \, dx$ **is a rational function.**
- **Rule: If** $e \neq c^2 d \wedge \left(p \in \mathbb{Z}^+ \vee p+\frac{1}{2} \in \mathbb{Z}^-\right)$**, let** $u = \int (d+e x^2)^p \, dx$**, then**

$$\int (d+e x^2)^p (a+b \operatorname{ArcSinh}[c x]) \, dx \longrightarrow u \, (a+b \operatorname{ArcSinh}[c x]) - b c \int \frac{u}{\sqrt{1+c^2 x^2}} \, dx$$

Program code:

```
Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[(d+e*x^2)^p,x]},
  Dist[a+b*ArcSinh[c*x],u,x] - b*c*Int[SimplifyIntegrand[u/Sqrt[1+c^2*x^2],x],x]] /;
FreeQ[{a,b,c,d,e},x] && NeQ[e,c^2*d] && (IGtQ[p,0] || ILtQ[p+1/2,0])
```

```
Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCosh[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[(d+e*x^2)^p,x]},
  Dist[a+b*ArcCosh[c*x],u,x] - b*c*Int[SimplifyIntegrand[u/(Sqrt[1+c*x]*Sqrt[-1+c*x]),x],x]] /;
FreeQ[{a,b,c,d,e},x] && NeQ[c^2*d+e,0] && (IGtQ[p,0] || ILtQ[p+1/2,0])
```

```
(* Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCosh[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[(d+e*x^2)^p,x]},
  Dist[a+b*ArcCosh[c*x],u,x] - b*c*Sqrt[1-c^2*x^2]/(Sqrt[1+c*x]*Sqrt[-1+c*x])*Int[SimplifyIntegrand[u/Sqrt[1-c^2*x^2],x],x]] /;
FreeQ[{a,b,c,d,e},x] && (IGtQ[p,0] || ILtQ[p+1/2,0]) *)
```

2: $\int (d+e x^2)^p (a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e \neq c^2 d \wedge p \in \mathbb{Z} \wedge (p > 0 \vee n \in \mathbb{Z}^+)$

Derivation: Algebraic expansion

Rule: If $e \neq c^2 d \wedge p \in \mathbb{Z} \wedge (p > 0 \vee n \in \mathbb{Z}^+)$, then

$$\int (d+e x^2)^p (a+b\,\text{ArcSinh}[c\,x])^n\,dx \longrightarrow \int (a+b\,\text{ArcSinh}[c\,x])^n\,\text{ExpandIntegrand}\left[(d+e x^2)^p,\,x\right]dx$$

Program code:

```
Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*ArcSinh[c*x])^n,(d+e*x^2)^p,x],x] /;
FreeQ[{a,b,c,d,e,n},x] && NeQ[e,c^2*d] && IntegerQ[p] && (p>0 || IGtQ[n,0])
```

```
Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*ArcCosh[c*x])^n,(d+e*x^2)^p,x],x] /;
FreeQ[{a,b,c,d,e,n},x] && NeQ[c^2*d+e,0] && IntegerQ[p] && (p>0 || IGtQ[n,0])
```

X: $\int (d+e x^2)^p (a+b\,\text{ArcSinh}[c\,x])^n\,dx$

Rule:

$$\int (d+e x^2)^p (a+b\,\text{ArcSinh}[c\,x])^n\,dx \longrightarrow \int (d+e x^2)^p (a+b\,\text{ArcSinh}[c\,x])^n\,dx$$

Program code:

```
Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  Unintegrable[(d+e*x^2)^p*(a+b*ArcSinh[c*x])^n,x] /;
FreeQ[{a,b,c,d,e,n,p},x]
```

```
Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  Unintegrable[(d+e*x^2)^p*(a+b*ArcCosh[c*x])^n,x] /;
FreeQ[{a,b,c,d,e,n,p},x] && IntegerQ[p]
```

```
Int[(d1_+e1_.*x_)^p_.*(d2_+e2_.*x_)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  Unintegrable[(d1+e1*x)^p*(d2+e2*x)^p*(a+b*ArcCosh[c*x])^n,x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,n,p},x]
```

Rules for integrands of the form $(d+e\,x)^p\,(f+g\,x)^q\,(a+b\,\text{ArcSinh}[c\,x])^n$

1: $\int (d+e\,x)^p\,(f+g\,x)^q\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e\,f+d\,g=0\,\bigwedge\,c^2\,d^2+e^2=0\,\bigwedge\,(p\mid q)\in\mathbb{Z}+\frac{1}{2}\,\bigwedge\,p-q\geq 0\,\bigwedge\,d>0\,\bigwedge\,\frac{g}{e}<0$

Derivation: Algebraic expansion

- **Basis: If** $e\,f+d\,g=0\,\bigwedge\,c^2\,d^2+e^2=0\,\bigwedge\,d>0\,\bigwedge\,\frac{g}{e}<0$**, then** $(d+e\,x)^p\,(f+g\,x)^q=\left(-\frac{d^2\,g}{e}\right)^q\,(d+e\,x)^{p-q}\,\left(1+c^2\,x^2\right)^q$
- **Rule: If** $e\,f+d\,g=0\,\bigwedge\,c^2\,d^2+e^2=0\,\bigwedge\,(p\mid q)\in\mathbb{Z}+\frac{1}{2}\,\bigwedge\,p-q\geq 0\,\bigwedge\,d>0\,\bigwedge\,\frac{g}{e}<0$**, then**

$$\int (d+e\,x)^p\,(f+g\,x)^q\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx\longrightarrow\left(-\frac{d^2\,g}{e}\right)^q\int (d+e\,x)^{p-q}\,\left(1+c^2\,x^2\right)^q\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$$

- **Program code:**

```
Int[(d_+e_.*x_)^p_*(f_+g_.*x_)^q_*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  (-d^2*g/e)^q*Int[(d+e*x)^(p-q)*(1+c^2*x^2)^q*(a+b*ArcSinh[c*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,g,n},x] && EqQ[e*f+d*g,0] && EqQ[c^2*d^2+e^2,0] && HalfIntegerQ[p,q] && GeQ[p-q,0] && GtQ[d,0] && LtQ[g/e,0]
```

2: $\int (d+e\,x)^p\,(f+g\,x)^q\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e\,f+d\,g==0\;\bigwedge\;c^2\,d^2+e^2==0\;\bigwedge\;(p\mid q)\in\mathbb{Z}+\frac{1}{2}\;\bigwedge\;p-q\geq 0\;\bigwedge\;\neg\left(d>0\;\bigwedge\;\frac{g}{e}<0\right)$

- **Derivation: Piecewise constant extraction**
- **Basis: If** $e\,f+d\,g==0\;\bigwedge\;c^2\,d^2+e^2==0$**, then** $\partial_x\,\frac{(d+e\,x)^q\,(f+g\,x)^q}{(1+c^2\,x^2)^q}==0$
- **Rule: If** $e\,f+d\,g==0\;\bigwedge\;c^2\,d^2+e^2==0\;\bigwedge\;(p\mid q)\in\mathbb{Z}+\frac{1}{2}\;\bigwedge\;p-q\geq 0\;\bigwedge\;\neg\left(d>0\;\bigwedge\;\frac{g}{e}<0\right)$**, then**

$$\int (d+e\,x)^p\,(f+g\,x)^q\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx\;\longrightarrow\;\frac{(d+e\,x)^q\,(f+g\,x)^q}{\left(1+c^2\,x^2\right)^q}\int (d+e\,x)^{p-q}\,\left(1+c^2\,x^2\right)^q\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$$

- **Program code:**

```
Int[(d_+e_.*x_)^p_*(f_+g_.*x_)^q_*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  (d+e*x)^q*(f+g*x)^q/(1+c^2*x^2)^q*Int[(d+e*x)^(p-q)*(1+c^2*x^2)^q*(a+b*ArcSinh[c*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,g,n},x] && EqQ[e*f+d*g,0] && EqQ[c^2*d^2+e^2,0] && HalfIntegerQ[p,q] && GeQ[p-q,0]
```

1: $\int (d+e\,x)^p\,(f+g\,x)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e\,f+d\,g==0\;\land\;c^2\,f^2+g^2==0\;\land\;p\notin\mathbb{Z}$

```
Int[(d_+e_.*x_^2)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (-d)^IntPart[p]*(d+e*x^2)^FracPart[p]/((1+c*x)^FracPart[p]*(-1+c*x)^FracPart[p])*
    Int[(1+c*x)^p*(-1+c*x)^p*(a+b*ArcCosh[c*x])^n,x] /;
FreeQ[{a,b,c,d,e,n,p},x] && EqQ[c^2*d+e,0] && Not[IntegerQ[p]]
```